

短 報

Polymodal な度数分布を
 正規分布へ分解する
 BASIC プログラム

赤 嶺 達郎¹⁾

A BASIC Program to Analyse
 the Polymodal Frequency
 Distribution into Normal
 Distributions.

TATSURO AKAMINE¹⁾

Abstract

The algorithm of this program is an iteration method which consists of the linear approximation and the method of least-squares. Variables are modified one after another. This program can be run on a micro computer, although it takes a longer time in comparison with larger computers.

Polymodal な度数分布を正規分布に分解することは、年齢形質が得られていない生物種を扱う場合等に用いられる基本的な手法である。大型計算機用のプログラムは既にいくつか発表されているが、利用できない研究者は田中 (1956) の方法等に依っている。

この BASIC プログラムは現在市販されている 10 万円以下の小型計算機を対象にしており、メモリーは 2 K バイトあれば十分である。算法は嶋津 (1980) に従い、1 次近似と最小 2 乗法である。メモリーを最少限におさえるためパラメータは 1 つずつ動かしている。したがって時間がかかることが欠点である

1) 〒951 新潟市水道町 1 丁目 5939-22
 日本海区水産研究所
 (Japan Sea Regional Fisheries Research Laboratory, Suido-cho, Niigata 951, Japan)

が、対象としている小型計算機の性能を考慮するとやむを得ない。

算 法

与えられた度数分布を F とし、階級の数 m 、階級の幅 h 、階級値の最小値 a 、最大値 b とする。分解する正規分布の数を n とおくと、求める式 f 、残差関数 d^2 は

$$f = \sum_{i=1}^n K_i \cdot N(\mu_i, \sigma_i, x) \quad \text{--- ①}$$

$$N = \frac{1}{\sqrt{2\pi}\sigma} \exp\left\{-\frac{(x-\mu)^2}{2\sigma^2}\right\}$$

$$d^2 = \sum_{x=a}^b dx^2 = \sum_x (F-f)^2 \quad \text{--- ②}$$

$$b = a + (m-1)h$$


となる (第 1 図)。① を 1 次近似すると

$$\Delta f \approx \sum_i \left\{ \frac{\partial f}{\partial K_i} \Delta K_i + \frac{\partial f}{\partial \mu_i} \Delta \mu_i + \frac{\partial f}{\partial \sigma_i} \Delta \sigma_i \right\}$$


$$\begin{cases} \frac{\partial f}{\partial K_i} = N(\mu_i, \sigma_i, x) \\ \frac{\partial f}{\partial \mu_i} = K_i \cdot N(\mu_i, \sigma_i, x) \cdot \frac{x - \mu_i}{\sigma_i^2} \\ \frac{\partial f}{\partial \sigma_i} = K_i \cdot N(\mu_i, \sigma_i, x) \cdot \frac{(x - \mu_i)^2 - \sigma_i^2}{\sigma_i^3} \end{cases}$$

となる。

ここで d^2 が最小となるように



の順でパラメータを変化させる。変化させるパラメータを α とおくと



第 1 図 変数の説明
 Fig. 1. Illustration of variables.

$$\Delta f \approx \frac{\partial f}{\partial \alpha} \Delta \alpha$$

②に代入して

$$d^2 = \sum_x \left\{ F - (f + \Delta f) \right\}^2 = \sum_x \left(d_x - \frac{\partial f}{\partial \alpha} \Delta \alpha \right)^2$$


ここで最小 2 乗法を用いると

$$\frac{\partial d^2}{\partial \Delta \alpha} = 0 \quad \text{より } \Delta \alpha = \frac{\sum_x \left(d_x \frac{\partial f}{\partial \alpha} \right)}{\sum_x \left(\frac{\partial f}{\partial \alpha} \right)^2}$$


となり、これが求める修正量である。

応 用 例

マイクロコンピュータ MB 6880 を用いて、田中 (1956) のキダイ体長組成の分解を行なった。 d^2 の収束状態を第 2 図に示す。結果を第 1 表に、X-Y



第 2 図 d^2 の収束状態
Fig. 2. Convergence of d^2 .



第 3 図 キダイの体長組成。ヒストグラムは測定値。曲線はこの BASIC プログラムより求められた正規分布とその和の曲線。

Fig. 3. Fork length frequency curve of porgy. The histogram shows the observed frequency, lines normal curves and the sum of each normal curves fitted by this BASIC program.

プロッターによる作図を第3図に示す。MB 6880では1回に平均8分ほどかかった。10回で打ち切った場合約80分かかることになる。

第1表 2方法による結果の比較
Table 1. Comparison of the results from two methods.

	1	2	3	4	5	d^2	
initial value	K	5000	4000	3000	1000	500	
	σ	11	15.5	20	24	27	
	μ	1	1	1.5	1.5	1.5	758923
value after 29 iterations	K	5676	4331	2703	798	492	
	σ	11.02	15.27	19.82	23.46	26.66	
	μ	0.823	1.162	1.478	1.202	1.475	6314
TANAKA'S graphical method	K	5627	4485	2630	840	461	
	μ	10.99	15.26	19.84	23.50	26.82	
	σ	0.8	1.2	1.4	1.2	1.4	19192

考 察

このプログラムでは最初に正規分布の数と初期値を入力する必要がある。よい初期値をモードのはつきりしたデータに入れるほど短い時間でできる。逆に、悪い初期値やモードのはつきりしないデータを入れた場合は、振動や発散が発生したり、停留点に入り込んでしまう可能性がある。このようなプログラムは客観性を得るために使用するものであり、入力や出力に関しては十分な考察を加える必要がある。

プログラムの改良点は

- 1) 入力, 出力を改良する。
 - 2) ループ脱出判断を改良する。
- 等が考えられる。

本論を終えるにあたり、X-Yプロッターによる作図や助言をしていただいた日本海区水産研究所資源部加藤史彦主任研究官に深謝します。

変数の説明
Correspondence of variables

- N : number of normal distributions
- A : minimum class mark
- H : class width
- M : number of classes
- D(1,J) : K_j F1 : f_x
- D(2,J) : μ_j D1 : d_x
- D(3, J) : σ_j D2 : d^2

$$G1 : \frac{\partial f}{\partial \alpha}$$

$$F(X) : F_x \quad A2 : \Delta \alpha$$

プログラムリスト (キダイのデータの入力例)
 Program list (DATA : an example of porgy)

```

10 REM POLYMODAL
20 REM シュリョク
30 READ N,A,M,H
40 PRINT 'N=';N
50 PRINT 'A=';A
60 PRINT 'M=';M
70 PRINT 'H=';H
80 DIM D(3,N),F(M)
90 STOP
100 FOR I=1 TO M
110 READ F(I):PRINT 'F(';I;')=';F(I)
120 NEXT I
130 STOP
140 FOR J=1 TO N
150 FOR I=1 TO 3
160 READ D(I,J):PRINT 'D(';I;',';J;')=';D(I,J)
170 NEXT I:NEXT J
180 STOP
190 READ D9:PRINT 'D9=';D9
200 STOP
210 REM ケイサン
220 D7=1
230 FOR I=1 TO 3
240 FOR J=1 TO N
250 D2=0:S1=0:S2=0
260 FOR K=1 TO M
270 X=A+(K-1)*H
280 F1=0
290 FOR L=1 TO N
300 P1=D(1,L):P2=X-D(2,L):P3=D(3,L)
310 F1=F1+P1*.398942/P3*EXP(-.5*Q2*Q2/P3/P3)
320 NEXT L
330 D1=F(K)-F1
340 Q1=D(1,J):Q2=X-D(2,J):Q3=D(3,J)
350 G1=-.398942/Q3*EXP(-.5*Q2*Q2/Q3/Q3)
360 ON I GOTO 390,370,380
370 G1=G1*G1*Q2/Q3/Q3:GOTO 390
380 G1=Q1*G1*(Q2*Q2-Q3*Q3)/Q3/Q3/Q3
390 D2=D2+D1*D1
400 S1=S1+D1*G1
410 S2=S2+G1*G1
420 NEXT K
430 A2=S1/S2
440 D(I,J)=D(I,J)+A2
450 PRINT
460 PRINT 'D7=';D7
470 PRINT 'I=';I,'J=';J
480 PRINT 'D2=';D2,'A2=';A2
490 NEXT J:NEXT I
500 D7=D7+1
510 IF D7<D9 GOTO 230
520 REM シュリョク
530 PRINT
540 FOR J=1 TO N
550 FOR I=1 TO 3
560 PRINT 'D(';I;',';J;')=';D(I,J)
570 NEXT I:NEXT J
580 END
1000 DATA 5,7.5,29,1
1010 DATA 7,79,509,2240,2341,623,476,1230,1439
1020 DATA 921,448,512,719,673,445,341,310,228
1030 DATA 168,140,114,64,22,0,2,2,0,0,1
1040 DATA 5000,11,1,4000,15.5,1,3000,20,1.5
1050 DATA 1000,24,1.5,500,27,1.5
1060 DATA 30
    
```

文 献

嶋津靖彦(1979). 体長組成から年齢組成を推定
する方法. 昭和54年度漁業資源研究会

議, 西日本底魚部会会議報告: 36—48.
田中昌一(1956). Polymodal な度数分布の
取扱方及びそのキダイ体長組成解析へ
の応用. 東海水研報告(14): 1—13.